フリーソフトを用いた数値計算　～ Scilab のすすめ

京都情報大学院大学　教授　作花 一志

はじめに

数値計算ソフトScilab のインストールについては [1] を，微分方程式の解法については [2] を参照されたい。今回この小文では，非線形方程式の解法と行列演算・固有値算出について述べる。

グラフ描画と方程式の解

$x^2-2x=0$ を解け

これは数学 I の範囲の易しい問題で中学生でも解ける子は多いだろう。因数分解でも解の公式を使っても，あるいは視察で（直観と言ってもよい）でもよい，もちろん解は $x=0$，2 である。ところがこんなに簡単に解ける方程式はきわめて少ない。２次方程式なら解の公式という便利なものがあるが，一般には解の公式があること自体が例外なのだ。高次方程式，分数方程式，無理方程式，三角方程式は何とかなっても対数方程式，指数方程式などとなるともうお手挙げである。解は存在するが，加減乗除のような普通の演算では解けないということだ。$x^2=2^x$では変数の置き換えなど解析的方法をいろいろやってみても解けない。むしろ直観に頼る方が賢明だ。小学生でも勘の鋭い子なら $x=2$ と答えるだろう。この場合，左辺の値も右辺の値も 4 である。その正解者のうちの何人かは $x=4$ のとき左辺の値も右辺の値も16であることを見抜くだろう。すなわち $x=2$，4 という解が存在する。では第３の解はないものだろうか。$x>0$ にはなさそうだが $x<0$ では? 直観もここまで，見方を変えよう。方程式 $f(x)=0$を解くということは $f(x)=0$ となる x を求める，すなわち

　$y=f(x)$ と x 軸（$y=0$）との交点を求める

ということ，言い換えれば逆変換 $x=f^{-1}(0)$ を行うということである。

図 1

そこで方程式の解法の第一段階は $y=f(x)$ のグラフを正確に描くということになる。最も簡単なグラフ描画の方法はブラウザを開き Google のサイトでただ単にその関数を検索するだけでよい。複数の関数を表示させたいときはカンマで並べる。図 1 は $y=x^2-2x$ と $y=2^x-x^2$ を描いたものである。$y=f(x)$ で一意的に定義される関数なら問題なく描ける。さらに $z=g(x,y)$ で表される立体図形も描画可能である。ただし媒介変数表示，極座標表示の式では使えない。ブラウザとしては Internet Explorer よりも Google Chrome のほうがよい。

代数関数，三角関数の場合は定義域が問題にならないが，指数関数，対数関数の場合には注意が必要である。

2^3（＝8）とは 2 を 3 回かけたものであり 2^5（＝32）とは 5 回の積である。また 2^{-2}とは $1/2^2$ のことすなわち指数がマイナスなら逆数をとればよい。では指数が分数，小数の場合はどうなるだろうか? $2^{1/2}$とは 2 乗して 2 になる数（すなわち 2 の平方根$\sqrt{2}$）と $2^{1/5}$とは 5 乗して 2 になる数（すなわち 2 の 5 乗根 $\sqrt[5]{2}$）と定義する。そうすれば $2^{3/5}=(2^{1/5})^3$ であるので一般に $2^{n/m}$ は 2 のm乗根の n 乗と定義できる。$2^{3.14}$ は $2^{314/100}$ より求められ，結局 2^x において x が有理数の場合はすべて計算できることになる。その数値自身は無理数で紙に書きとめることはできないが，確かに実在する数である。

では x が無理数の場合は? 2^π，$2^{\sqrt{2}}$ などは実在するのだろうか?

π に収束する次のような無限数列を考えよう。

　$a_n=\{3,\ 3.14\quad 3.141\quad 3.1415\quad 3.14159\quad 3.141592\cdots\}$

これに対して次のような数列

　$b_n=\{2^3,\ 2^{3.14}\quad 2^{3.141}\quad 2^{3.1415}\quad 2^{3.14159}\quad 2^{3.141592}\cdots\}$

は有理数であるが

　$n\to\infty$において　$a_n\to\pi$　$b_n\to 2^\pi$ となる（厳密には証明が要るが）。

このように考えればすべての実数 x において 2^xは計算できることになった。指数関数 $y=a^x$ は $a>1$ なら単調増加，$0<a<1$ なら単調減少，そして $a=1$，$a=0$ なら一定である。では $a<0$ の場合はどうであろうか? 底がマイナスの場合の指数関数はどう定義したらいいだろうか?

$f(x)=(-2)^x$ の場合を考えてみよう。この関数の値はxが整数の場合は実数値だが，そうでなければ複素数の値になる。その値を求めるには Scilab を起動して x=1.52；(−2)^x とすれば 0.1800776 −2.8622513 i と複素数で出力される。では $y=f(x)$ のグラフを描いてみる。図 2 では実部を濃で，虚部を

淡でプロットしてある。$y=(-2)^x$ を直接プロットすると実部と一致している。



図 2

```
// 関数定義
deff('y=f1(x)','y=real((-2)^x)')
deff('y=f2(x)','y=imag((-2)^x)')
// 範囲、格子
x=[-3:0.1:3];
xgrid()
// プロット
plot(x,f1,'r')
plot(x,f2,'g')
```

グラフを見ると x が整数のとき虚部の値は 0，半整数のとき実部の値は 0 となっているようにみえる。

　$f(1)=-2$　　$f(2)=4$

　$f(1.5)=(-2)^{1.5}=-2\sqrt{-2}=-2\sqrt{2}\,i$

　$f(-0.5)=(-2)^{-0.5}=1／(\sqrt{-2})=-\sqrt{2}\,i／2$

である。

-2 を $r(\cos\theta+i\sin\theta)$ の形で複素数表示してみよう。

r 絶対値，θ 偏角。

　実数部分を比較　$-2=r\cos\theta$

　虚数部分を比較　　$0=r\sin\theta$

よって $\theta=\pi$，$r=2$ であり

$y=(-2)^x$ の実部，虚部はそれぞれ $2^x(\cos(x\pi))$，$2^x(\sin(x\pi))$ となる。よって

x が整数のとき　$\cos(x\pi)=\pm1$　$\sin(x\pi)=0$　だから y は実数

x が半整数のとき　$\cos(x\pi)=0$　$\sin(x\pi)=\pm1$　だから y は純虚数

同様にして $y=x^x$ のグラフを実部虚部それぞれ描いてみる（図 3）。

$x\geqq0$ または $x<0$ で整数のとき　虚部 $=0$ すなわち y は実数。

　$(-2)^{-2}=1／4$

$x<0$ で半整数のとき　実部 $=0$ すなわち y は純虚数 。

　$(-1.5)^{-1.5}=2\sqrt{6}\,i／9$



図 3

方程式の数値的解法ではまずグラフを描き x 軸との交点を求めるわけだが，その際，適当な初期値から出発して何回かの繰り返し演算が必要だ。初期値は視察的に決めるが，繰り返し演算には収束判定条件が難しく従来の数値計算プログラミングはここで足踏みすることが多かった。しかし Scilab では fsolve という関数を利用すれば容易に解ける。

例えば $exp(x)-x=2$ を解くとき，まずグラフを描いて，初期値として 1 をとれば 1.1461932 が得られる。また初期値を -1 とすれば -1.8414057 が得られる。

初期値を 0 とすると解は 0.0000 と表示されるが，この場合 $f'(0)=0$ となるから交点は求まらず解なしである。



図 4

```
// 方程式定義
deff('y=f(x)','y=exp(x)-x-2');
// タイトル
 xset("font size",3);
xlfont("Century",16);
xstring(-1,3, 'exp(x)-x-2=0');

// グラフ描画
xgrid()
x=-5:0.05:2;
plot(x,f);
plot(x,0);

//  初期値
p=[" x0 =  "];
v=['-1'];
as=x_mdialog(" 初期値を入力 ",p,v);
x0=msscanf(as(1),'%f');

// 解
 sol=fsolve(x0,f);
//  出力
 mprintf(" 解 = %f \n",sol);
 xstring(sol-2,2," 解 =" + string(sol));
plot2d(sol,0,style=-4)
```

グラフを描き初期値を入力し解を求めるScilabプログラミングは左のようになる。なお初期値を入力するダイアログは x_mdialog を使う。

図 5

ではここで方程式

$2^x=x^2$

に戻る。

$f(x)=2^x-x^2$

y=f(x) のグラフは確かに（2，0）および（4，0）を通るが，x＝−1 あたりにも交点は存在することがわかる。その値は解析的には求まらないが，初期値を−1とすれば近似的に−0.766685となる（図 6）。



図 6

次に $x^x=1$ を解いてみよう。両辺の対数を取れば xlog(x)=0 だから x＝0 または log(x)=0 すなわち解は x＝0, 1 であることは解析的にわかるが他に解はないものか？

$f(x)=x^x-1$ のグラフを描くと x<0 に解がありそうなので初期値を−0.5 とすれば x ＝−0.25 と求まる。ところが x＝−0.25 において x^x の値は 1+i であり，実部が 1 となる x の値を求めたことになる（図 7）。

なお x=0 は f(x) の定義域ではないが x →0 の極限で f(x)→0 となることは解析的に証明されているので解に含める。



図 7

行列演算

行列演算は Excel でもできることが多いが，Scilab では固有値・固有ベクトルまでも簡単に求められる。ここでは 2 次の正方行列を例にとる。

行列 $A=\begin{pmatrix} 1 & 4 \\ 2 & 3 \end{pmatrix}$ は A=[1,4 ;2,3] と書く。セミコロンは改行を意味する。

A(1,2)＝ 4 は第 1 行第 2 列成分を，A(1,:) は第 1 行を，A(:,1) は第 1 列を表す。

ベクトルは v=[1;3] のように 2 行 1 列の行列とみなして扱う。

2 つの行列の和・差・積はそのまま A+B，A-B，A*B とすれば計算される。転置は A'，逆行列 A^{-1} は inv(A)，行列式｜A｜は det(A) で求まる。

$A^{-1}=\begin{pmatrix} -0.6 & 0.8 \\ 0.4 & -0.2 \end{pmatrix}$　　｜A｜= −5

連立一次方程式

$x+4y=3$

$2x+3y=5$

は Ax=b と表されこれを解くには行列 A とベクトル b の成分を与えて $x=A^{-1}b$ を計算すればよい。

b=[3;5] であるから A¥b または inv(A)*b とすれば x=2.2, y=0.2 と求まる。ただし det(A)=0 の場合は適用できない。

lamda＝spec(A) より A の固有値−1，5 が

[V,D]=spec(A) とすれば 2 つの行列 V と D が得られる。

$$V = \begin{pmatrix} -0.8944272 & -0.7071068 \\ 0.4472136 & -0.7071068 \end{pmatrix}$$

の第 1 列は固有値－1 に対応する固有ベクトルで，第 2 列は固有値 5 に対応する固有ベクトルである。

また

$$D = \begin{pmatrix} -1. & 0 \\ 0 & 5. \end{pmatrix}$$

は固有値を対角成分とする対角行列であるが，inv(V)*A*V＝D すなわち A は V によって対角化されている。

A=[1,4;－2,3] のとき固有値は複素数となり，行列 V や D の成分も複素数となる。

計算結果の出力は editvar 行列名 とすれば表形式で出力されて見やすくなる。図 8 では固有ベクトルを各列とする V が表記されている。3 次以上の場合でも同様に計算できる。

```
A=[1,4;2,3];  editvar A;
A1=A(1,:); editvar A1;
B=[-1,5;-2,3]; editvar B;
WA=A+B; editvar WA;
SA=A-B; editvar SA;
AB=A*B; editvar AB;
BA=B*A; editvar BA;
TA=A'; editvar TA;
IA=inv(A) ;editvar IA;
DA=det(A);editvar DA;
b=[3;5]; editvar b;
x=IA*b; editvar x;
lamda=spec(A);editvar lamda;
[V,D]=spec(A);editvar V;editvar D
```



図 8

一次変換の演算は行列とベクトルの積でできる。ベクトル v を行列 A により 60 度回転，行列 B により x 軸で反転しかつ 2 倍拡大してみよう。A による変換の結果のベクトルは濃で B による変換の結果のベクトルは淡で図示された（図 9）。

```
// 回転角度
angle=60*%pi/180;
c=cos(angle);s=sin(angle);
//　その行列
A=[c,-s;s,c]
//x 軸で反転　2 倍拡大
B=[2,0;0,-2]
//　ベクトル　v
v=[1;2]
vx=[0;v(1)];
vy=[0;v(2)];
//　一次変換
va=A*v
vax=[0;va(1)];
vay=[0;va(2)];
vb=B*v
vbx=[0;vb(1)];
vby=[0;vb(2)];

// 図示
xset("thickness",3);
plot2d4(vx,vy,1,rect=[-5,-5,5,5],frameflag=3)
plot2d4(vax,vay,2,rect=[-5,-5,5,5],frameflag=3)
plot2d4(vbx,vby,5,rect=[-5,-5,5,5],frameflag=3)
xset("thickness",1)
xgrid()
```

言うまでもないことだが，｜A｜は一次変換における面積（3 次元なら体積）拡大率を表し，A^{-1} は逆変換を表す。また，plot2d4 でベクトルを矢印で表すことができる。

固有値・固有ベクトルの解説は通常，線形代数の最後の章で行われ，その応用は力学から心理学まで多方面にわたっている。オーソドックスな計算法を理解するのは多大な時間と忍耐を要し，さまざまな技法が提案されているが，最も容易にかつ安価に理解できるのは Scilab を使うことだろう。筆者は現在「ビジネス統計学特論」の主成分分析の解説の際にこの方法を使っている。



図 9

［参考文献］

[1] http://www.scilab.org/

[2] http://web1.kcg.edu/~sakka/num/scilab/dfe.pdf

作花 一志　**Kazuyuki Sakka**

京都情報大学院大学教授。

京都大学理学士，同大学院博士課程修了（宇宙物理学専攻），理学博士。

研究分野は計算天文学（惑星会合検索，小惑星データベースなど），視覚的数学理科教育法（マルチメディア教材開発）。

元京都コンピュータ学院鴨川校校長，元天文教育普及研究会編集委員長。